


大なわとびさっちゃんを 作ろう

(日本語)



スペースキーでジャンプして大なわとびをする
IchigoJamのゲームプログラムです




IchigoJamのずのう(ファームウェア)を
バージョン1.4Iにして作ろう！




IchigoJam webでも作れるよ！
fukuno.jig.jp/app/IchigoJam/

(1)さいしょのせってい・
さっちゃんをひょうじ



入力できたら、F5でじっこう



さっちゃんをひょうじ

(2) なわをひょうじ

Shift+れ ↓ め ↓

```
40 X=30+COS(T)/12
```


 ↑ ↑

 Shift+8 Shift+9

```
50 Y=22+SIN(T)/12
```

```
60 DRAW X,Y
```

入力できたら、F5でじっこう



(3)なわを回す


オー ↓ ゼロ ↓ オー ↓ ゼロ ↓ ゼロ ↓

```
20  FOR T=-260 TO 90 STEP 10
```

ほ ↑

```
120 NEXT
```

入力できたら、F5でじっこう




なわが1しゅう回る

(4) なわのあとをけす

```
30 DRAW X,Y,0
```

入力できたら、F5でじっこう



あとがのこらずに、1しゅう回る

(5) なわ回しをくりかえす

```
130 GOTO 20
```




オー



ゼロ

入力できたら、F5でじっこう



なわが回りつづける

(6) さっちゃんをジャンプさせる

```
80 IF J Z=Z+V:V=V+1
```


Shift+1

Shift+6



```
110 IF !J && INKEY()=32  
J=1:V=-4
```

入力できたら、F5でじっこう




スペースキーをおすと、さっちゃんがジャンプ
ただし、ちやくちせずに、そのままおちていく

(7) さっちゃんをちやくちさせる

```
100 IF V=5 J=0
```

入力できたら、F5でじっこう




スペースキーをおすと、さっちゃんがジャンプしてちやくちする

(8) さっちゃんのあとをけす

```
70 LC 15,Z:?" "
```

入力できたら、F5でじっこう




スペースキーをおして、さっちゃんがジャンプしても
あとがのこらなくなる

(9) なわをふんだかをはんてい

```
130 IF Z!=20 S=S+1:GOTO 20
```

入力できたら、F5でじっこう




なわをふむと、しゅうりょうする

(10)とんだかいすうをひょうじ

140 ?S

入力できたら、F5でじっこう



なわをふむと、とんだかいすうをひょうじ

●大なわとびさっちゃん かんせい

10 CLV:CLS:Z=20	初期設定
20 FOR T=-260 TO 90 STEP 10	なわを回す
30 DRAW X,Y,0	なわを消す
40 X=30+COS(T)/12	なわのX座標を計算
50 Y=22+SIN(T)/12	なわのY座標を計算
60 DRAW X,Y	なわを表示
70 LC 15,Z:?" "	さっちゃんを消す
80 IF J Z=Z+V:V=V+1	ジャンプしていたら、座標を変える
90 LC 15,Z:?"0"	さっちゃんを表示
100 IF V=5 J=0	着地したらジャンプを止める
110 IF !J && INKEY()=32 J=1:V=-4	スペースキーが押されたらジャンプ
120 NEXT	くりかえし
130 IF Z!=20 S=S+1:GOTO 20	なわをふんでいなければもどる
140 ?S	とんだ回数を表示して終了

なんかいとべるか
ちょうせんしよう